Complex Contours and Contour Integrals

Mth 611 Spring 2002

May 31 2002 - Maple 6
Revised June 1, 2002
Bent E. Petersen
Filename: 611s2002_complex_contours.mws

```
> restart; with(plots):
Warning, the name changecoords has been redefined
```

Here's a little routine to plot curves in the complex plane:

```
> cplot:=(z,R)->plot([Re(z),Im(z),R],args[3..nargs]):
```

Here z is an expression (not a function) in some real variable, say t, and R is a range of the form t=a..b Note you can pass any number of variables to a Maple routine, so we make a provision for extra variables by picking them up from the args list and then passing them to the plot command.

One can also simply use Maple's built-in command **complexplot()**. It is very flexible and faster than cplot(). However, Maple has many built-in commands. Sometimes it is difficult and time consuming to find the one you want. Therefore it is useful to have a little experience at devising rough homebrew solutions.

Let's test our routine by plotting a circle.

```
> cplot(exp(I*t),t=0..2*Pi,color=blue,thickness=3);
```



Here's the same plot as produced by complexplot().

```
> complexplot(exp(I*t),t=0..2*Pi,color=blue,thickness=3);
```



It may not look like a circle because the axes are scaled differently. We can force Maple to use the same scale on each axis (but it is hardly ever worth doing so). Here's that circle again -

```
> cplot(exp(I*t),t=0..2*Pi,color=blue,thickness=3,scaling=constraine
  d);
```



Recall the number of solutions of f(w)=z in the disk of radius r (for example) is given by the winding number about z of the image under f of the boundary of the disk. Here's an example:

```
> eqn:=z=w^3+w^2;
```

$$eqn := z = w^3 + w^2$$

```
> h:=subs(w=r*exp(I*t),rhs(eqn)):  `h(t)`=simplify(%);
```

$$h(t) = r^3 \mathbf{e}^{(3\,I\,t)} + r^2 \mathbf{e}^{(2\,I\,t)}$$

```
> cplot(subs(r=2,h),t=0..2*Pi,color=brown,thickness=3,title="w^3+w^2
  , r=2");
```



We see if z lies inside the small loop then f(w)=z has 3 solutions (the winding number) in the disk with radius 2 and center at the origin, D(0,2).

It is not necessary to specify thickness or color. The default is thickness=1 and color=red (at least in an unmodified Maple 6).

```
> cplot(subs(r=1,h),t=0..2*Pi,title="w^3+w^2, r=1");
```



Note f(w) has a critical point at 0 and at -2/3. The second derivative is nonzero at these points so it follows that f(w) is 2-to-1 in a punctured neighborhood of each critical point. The next plot shows clearly that for any z we have at most 2 solutions of f(w)=z in the disk D(0,3/5).

```
> cplot(subs(r=3/5,h),t=0..2*Pi,title="w^3+w^2, r=3/5");
```



Now let's create a little routine to compute contour integrals - actually we will define two routines, one inert. Note we assume we are dealing with differentiable contours.

```
> cint:=proc(beta,R,f,z)
>   local  t,g; t:=lhs(R);
>   g:=subs(z=beta,f)*diff('beta',t);
>   int(g,R,args[5..nargs]);
> end:
>
> Cint:=proc(beta,R,f,z)
>   local  t,g; t:=lhs(R);
>   g:=subs(z=beta,f)*diff('beta',t);
>   Int(g,R,args[5..nargs]);
> end:
```

Here beta is the contour (an expression in a real variable, not a function), R is a range for the parameter in beta, f is an expression in a complex variable and z is that variable.

Note we have used deferred evaluation of beta in the differentiation command. This prevents the int() command from chocking on booleans (as would occur in dealing with piecewise smooth curves).

```
> 1/(2*Pi*I)*cint(subs(r=2,h),t=0..2*Pi,1/z,z);
```

That is certainly the winding number of the contour "w^3+w^2, r=2" about about the origin. Let try the winding number about 10.

If you try cint() it will take a long time and you will get quite a mess. We use the inert Cint() instead and then evalf() (evaluate as floating point). The evalf() call after an inert integral call alerts Maple that we want to compute the integral numerically rather than symbolically. Maple uses its default numerical method though we could specify any of a number of methods. Skipping the attempt to do a symbolic evaluation can save a substantial amount of time.

```
> 1/(2*Pi*I)*Cint(subs(r=2,h),t=0..2*Pi,1/(z-10),z): evalf(%,8);
  evalf(%%,10);
```

$$1.0000000 + .47746482\ 10^{-9}\ I$$

$$.9999999995 - .9549296585\ 10^{-12}\ I$$

It is pretty clear the imaginary part here is due to roundoff and the winding number is 1 (as expected).

Note cint() can also be used for nonclosed contours:

```
> h2:=1-t;
```

$$h2 := 1 - t$$

```
> cint(h2,t=0..1,1/(w-z),w);
```

$$\ln\left(\frac{z}{-1+z}\right)$$

Now let's consider a piecewise smooth contour - a square:

```
> h3a:=(1-t)*(-1-I)+t*(1-I):
> h3b:=(1-t)*(1-I)+t*(1+I):
> h3c:=(1-t)*(1+I)+t*(-1+I):
> h3d:=(1-t)*(-1+I)+t*(-1-I):
> h3:=piecewise(t<1,h3a,t<2,subs(t=t-1,h3b),t<3,subs(t=t-2,h3c),t<=4
  ,subs(t=t-3,h3d));
```

$$h3 := \begin{cases} -(1+I)(1-t)+(1-I)\,t & t<1 \\ (1-I)(2-t)+(1+I)(t-1) & t<2 \\ (1+I)(3-t)-(1-I)(t-2) & t<3 \\ -(1-I)(4-t)-(1+I)(t-3) & t\le 4 \end{cases}$$

Note we were careful at t=4. Otherwise Maple will consider h3(4) to be 0 and cplot() will produce a line segment joining the origin to one vertex of our square.

```
> cplot(h3,t=0..4,thickness=2,color=brown);
```



Now let's consider a contour integral over the square:

```
> g:=z/sin(z);
```

$$g := \frac{z}{\sin(z)}$$

```
> Cint(h3,t=0..4,g,z): evalf(%); evalf(%%,16);
```

$$-.1\ 10^{-11} + 0.\ I$$

$$0. + 0.\ I$$

We expected to get 0 by Cauchy's theorem since z/sin(z) has a removeable singularity at the origin. The default precision in Maple is 10 decimal digits. The nonzero value we get a first is due to round-off.

```
>
```